
1. [bookmark: _Toc164099173]Change 1
	Function name
	LPP_Templates.ttcn

	Reason for change
	1. Based on CR R5-241577, GNSS-AuxiliaryInformation should also be provided in case of BDS also. Currently passing ‘omit’ in the template cs_GenericAssistDataReqElement_BDS for GNSS-AuxiliaryInformation while requesting GNSS-GenericAssistData depending on supported GNSSs.
2. Required code missing to handle and building IE GNSS_ID_BDS_r16 for all satellites in the list. Please refer Prose CR R5-241577
3. Currently the function fl_Build_cs_GNSS_AuxiliaryInformation does not provide support to build IE GNSS_AuxiliaryInformation for all Satellites in the list in case of BDS

	Summary of change
	1. Removed ‘omit’ for GNSS-AuxiliaryInformation from the template cs_GenericAssistDataReqElement_BDS as it is needed.
2. Added a function and template to set IE GNSS_ID_BDS_r16 for all satellites in the list in case of BDS.
3. Added support in the function fl_Build_cs_GNSS_AuxiliaryInformation to build IE GNSS_AuxiliaryInformation for BDS as well.

	TTCN module
	LPP_Templates.ttcn

	MCC160 Comment
	



Before change
<<SKIPPED CODE>>

template (value) GNSS_GenericAssistDataReqElement cs_GenericAssistDataReqElement_BDS :=
    /* @status    APPROVED (POS) */
    cs_GNSS_GenericAssistDataReqElement(cs_GNSS_ID_Bds,
                                        cs_GNSS_TimeModelListReq(cs_GNSS_TimeModelElementReq(1, true)), // GPS
                                        cs_GNSS_NavigationModelReq_ReqNavList(tsc_SvID_1_2_7_18_21_27, false),
                                        cs_GNSS_AcquisitionAssistanceReq(0),
                                        cs_GNSS_AlmanacReq,
                                        -,
                                        omit);


<<SKIPPED CODE>>

// templates for GNSS-AuxiliaryInformation:
  
  template (value) GNSS_ID_GPS_SatElement cs_GNSS_ID_GPS_SatElement(integer p_SatelliteID,
                                                                    GNSS_ID_GPS_SatElement p_ProvidedSatElement) :=
  { /* @status    APPROVED (POS) */
    svID := cs_SV_ID(p_SatelliteID),
    signalsAvailable := p_ProvidedSatElement.signalsAvailable
  };

  
  /*
   * @desc      Builds IE GNSS_ID_GPS for all satellites in the list
   * @param     p_SatelliteIdList
   * @param     p_ProvidedGNSS_ID_GPS
   * @return    template (value) GNSS_ID_GPS
   * @status    APPROVED (POS)
   */
  function fl_Build_cs_GNSS_ID_GPS(IntegerList_Type p_SatelliteIdList,
                                   GNSS_ID_GPS p_ProvidedGNSS_ID_GPS) return template (value) GNSS_ID_GPS
  {
    var template (value) GNSS_ID_GPS v_GNSS_ID_GPS;
    var integer i;
    for (i := 0; i < 6; i:= i + 1) {
      v_GNSS_ID_GPS[i] := cs_GNSS_ID_GPS_SatElement(p_SatelliteIdList[i], p_ProvidedGNSS_ID_GPS[i]);
    }
    return v_GNSS_ID_GPS;
  }

  template (value) GNSS_ID_GLONASS_SatElement cs_GNSS_ID_GLONASS_SatElement(integer p_SatelliteID,
                                                                            GNSS_ID_GLONASS_SatElement p_ProvidedSatElement) :=
  { /* @status    APPROVED (POS) */
    svID := cs_SV_ID(p_SatelliteID),
    signalsAvailable := p_ProvidedSatElement.signalsAvailable,
    channelNumber := p_ProvidedSatElement.channelNumber
  };

  /*
   * @desc      Builds IE GNSS_ID_GLONASS for all satellites in the list
   * @param     p_SatelliteIdList
   * @param     p_ProvidedGNSS_ID_GLONASS
   * @return    template (value) GNSS_ID_GLONASS
   * @status    APPROVED (POS)
   */
  function fl_Build_cs_GNSS_ID_GLONASS(IntegerList_Type p_SatelliteIdList,
                                       GNSS_ID_GLONASS p_ProvidedGNSS_ID_GLONASS) return template (value) GNSS_ID_GLONASS
  {
    var template (value) GNSS_ID_GLONASS v_GNSS_ID_GLONASS;
    var integer i;
    for (i := 0; i < 6; i:= i + 1) {
      v_GNSS_ID_GLONASS[i] := cs_GNSS_ID_GLONASS_SatElement(p_SatelliteIdList[i], p_ProvidedGNSS_ID_GLONASS[i]);
    }
    return v_GNSS_ID_GLONASS;
  }

  /*
   * @desc      Builds the IE GNSS_AuxiliaryInformation for all Satellites in the list
   * @param     p_SatelliteIdList
   * @param     p_ProvidedGNSS_AuxiliaryInformation
   * @return    template (value) GNSS_AuxiliaryInformation
   * @status    APPROVED (POS)
   */
  function fl_Build_cs_GNSS_AuxiliaryInformation(IntegerList_Type p_SatelliteIdList,
                                                 GNSS_AuxiliaryInformation p_ProvidedGNSS_AuxiliaryInformation) return template (value) GNSS_AuxiliaryInformation
  {
    var template (value) GNSS_AuxiliaryInformation v_GNSS_AuxiliaryInformation;

    if (ischosen(p_ProvidedGNSS_AuxiliaryInformation.gnss_ID_GPS)) {
      v_GNSS_AuxiliaryInformation.gnss_ID_GPS := fl_Build_cs_GNSS_ID_GPS(p_SatelliteIdList, p_ProvidedGNSS_AuxiliaryInformation.gnss_ID_GPS);
    }
    else {
      v_GNSS_AuxiliaryInformation.gnss_ID_GLONASS := fl_Build_cs_GNSS_ID_GLONASS(p_SatelliteIdList, p_ProvidedGNSS_AuxiliaryInformation.gnss_ID_GLONASS);
    }
    return v_GNSS_AuxiliaryInformation;
  }

<<SKIPPED CODE>>

After change
<<SKIPPED CODE>>

template (value) GNSS_GenericAssistDataReqElement cs_GenericAssistDataReqElement_BDS :=
    /* @status    APPROVED (POS) */
    cs_GNSS_GenericAssistDataReqElement(cs_GNSS_ID_Bds,
                                        cs_GNSS_TimeModelListReq(cs_GNSS_TimeModelElementReq(1, true)), // GPS
                                        cs_GNSS_NavigationModelReq_ReqNavList(tsc_SvID_1_2_7_18_21_27, false),
                                        cs_GNSS_AcquisitionAssistanceReq(0),
                                        cs_GNSS_AlmanacReq,
                                        -);

<<SKIPPED CODE>>

// templates for GNSS-AuxiliaryInformation:
  
  template (value) GNSS_ID_GPS_SatElement cs_GNSS_ID_GPS_SatElement(integer p_SatelliteID,
                                                                    GNSS_ID_GPS_SatElement p_ProvidedSatElement) :=
  { /* @status    APPROVED (POS) */
    svID := cs_SV_ID(p_SatelliteID),
    signalsAvailable := p_ProvidedSatElement.signalsAvailable
  };

  
  /*
   * @desc      Builds IE GNSS_ID_GPS for all satellites in the list
   * @param     p_SatelliteIdList
   * @param     p_ProvidedGNSS_ID_GPS
   * @return    template (value) GNSS_ID_GPS
   * @status    APPROVED (POS)
   */
  function fl_Build_cs_GNSS_ID_GPS(IntegerList_Type p_SatelliteIdList,
                                   GNSS_ID_GPS p_ProvidedGNSS_ID_GPS) return template (value) GNSS_ID_GPS
  {
    var template (value) GNSS_ID_GPS v_GNSS_ID_GPS;
    var integer i;
    for (i := 0; i < 6; i:= i + 1) {
      v_GNSS_ID_GPS[i] := cs_GNSS_ID_GPS_SatElement(p_SatelliteIdList[i], p_ProvidedGNSS_ID_GPS[i]);
    }
    return v_GNSS_ID_GPS;
  }

  template (value) GNSS_ID_GLONASS_SatElement cs_GNSS_ID_GLONASS_SatElement(integer p_SatelliteID,
                                                                            GNSS_ID_GLONASS_SatElement p_ProvidedSatElement) :=
  { /* @status    APPROVED (POS) */
    svID := cs_SV_ID(p_SatelliteID),
    signalsAvailable := p_ProvidedSatElement.signalsAvailable,
    channelNumber := p_ProvidedSatElement.channelNumber
  };

  /*
   * @desc      Builds IE GNSS_ID_GLONASS for all satellites in the list
   * @param     p_SatelliteIdList
   * @param     p_ProvidedGNSS_ID_GLONASS
   * @return    template (value) GNSS_ID_GLONASS
   * @status    APPROVED (POS)
   */
  function fl_Build_cs_GNSS_ID_GLONASS(IntegerList_Type p_SatelliteIdList,
                                       GNSS_ID_GLONASS p_ProvidedGNSS_ID_GLONASS) return template (value) GNSS_ID_GLONASS
  {
    var template (value) GNSS_ID_GLONASS v_GNSS_ID_GLONASS;
    var integer i;
    for (i := 0; i < 6; i:= i + 1) {
      v_GNSS_ID_GLONASS[i] := cs_GNSS_ID_GLONASS_SatElement(p_SatelliteIdList[i], p_ProvidedGNSS_ID_GLONASS[i]);
    }
    return v_GNSS_ID_GLONASS;
  }
  
 template (value) GNSS_ID_BDS_SatElement_r16  cs_GNSS_ID_BDS_SatElement(integer p_SatelliteID,
                                                                    GNSS_ID_BDS_SatElement_r16 p_ProvidedSatElement) :=
  {
    svID_r16 := cs_SV_ID(p_SatelliteID),
    satType_r16 := p_ProvidedSatElement.satType_r16
  };

  
  function fl_Build_cs_GNSS_ID_BDS_r16(IntegerList_Type p_SatelliteIdList,
                                   GNSS_ID_BDS_r16 p_ProvidedGNSS_ID_BDS) return template (value) GNSS_ID_BDS_r16
  {
    var template (value) GNSS_ID_BDS_r16 v_GNSS_ID_BDS;
    var integer i;
    for (i := 0; i < 6; i:= i + 1) {
      v_GNSS_ID_BDS[i] := cs_GNSS_ID_BDS_SatElement(p_SatelliteIdList[i], p_ProvidedGNSS_ID_BDS[i]);
    }
    return v_GNSS_ID_BDS;
  }


  /*
   * @desc      Builds the IE GNSS_AuxiliaryInformation for all Satellites in the list
   * @param     p_SatelliteIdList
   * @param     p_ProvidedGNSS_AuxiliaryInformation
   * @return    template (value) GNSS_AuxiliaryInformation
   * @status    APPROVED (POS)
   */
  function fl_Build_cs_GNSS_AuxiliaryInformation(IntegerList_Type p_SatelliteIdList,
                                                 GNSS_AuxiliaryInformation p_ProvidedGNSS_AuxiliaryInformation) return template (value) GNSS_AuxiliaryInformation
  {
    var template (value) GNSS_AuxiliaryInformation v_GNSS_AuxiliaryInformation;

    if (ischosen(p_ProvidedGNSS_AuxiliaryInformation.gnss_ID_GPS)) {
      v_GNSS_AuxiliaryInformation.gnss_ID_GPS := fl_Build_cs_GNSS_ID_GPS(p_SatelliteIdList, p_ProvidedGNSS_AuxiliaryInformation.gnss_ID_GPS);
    }
    else if (ischosen(p_ProvidedGNSS_AuxiliaryInformation.gnss_ID_GLONASS)){
      v_GNSS_AuxiliaryInformation.gnss_ID_GLONASS := fl_Build_cs_GNSS_ID_GLONASS(p_SatelliteIdList, p_ProvidedGNSS_AuxiliaryInformation.gnss_ID_GLONASS);
    }
    else{
      v_GNSS_AuxiliaryInformation.gnss_ID_BDS_r16 := fl_Build_cs_GNSS_ID_BDS_r16(p_SatelliteIdList, p_ProvidedGNSS_AuxiliaryInformation.gnss_ID_BDS_r16);
    }
    return v_GNSS_AuxiliaryInformation;
  }


<<SKIPPED CODE>>



1. [bookmark: _Toc164099174]Change 2
	Function name
	f_POS_SelectTemplate_GenericAssistDataElement_BDS()

	Reason for change
	This SatelliteIdList simulated for Glonass so incorrect SatelliteIdList is using for BDS

	Summary of change
	Corrected SatelliteIdList in the function f_POS_SelectTemplate_GenericAssistDataElement_BDS

	TTCN module
	CommonPOS_Functions.ttcn

	MCC160 Comment
	



Before change
function f_POS_SelectTemplate_GenericAssistDataElement_BDS(GNSS_GenericAssistData p_RetrievedGenericAssistData,
                                                             boolean p_UEB) return template (value) GNSS_GenericAssistDataElement
  {
    var template (value) GNSS_GenericAssistDataElement v_GenericAssistDataElement;
    var integer i := f_POS_FindIndexForGNSSMethod(p_RetrievedGenericAssistData, cs_GNSS_ID_Bds); //Indicates the position of the BDS data in the retrieved data
    
    if (i == -1) {//BDS element was not found in the list
      FatalError (__FILE__, __LINE__, "Invalid CellId"); // This function is only called if PICS indicate BDS support. If BDS assistance data is not there, it is a FatalError
    }
    else {
      //Include all possible elements in the template
      // @sic R5-240554 sic@
      v_GenericAssistDataElement := cs_GNSS_GenericAssistDataElement(cs_GNSS_ID_Bds,
                                                                     cs_GNSS_TimeModelList(tsc_GNSS_TO_ID_GPS, p_RetrievedGenericAssistData[i].gnss_TimeModels[0]),
                                                                     cs_NavigationModel_Bds(p_RetrievedGenericAssistData[i].gnss_NavigationModel),
                                                                     cs_GNSS_AcquisitionAssistance(cs_GNSS_SignalID(fl_DetermineBDS_SignalID_GNSS_AcquAssist()), // Beidou B1I or B1C
                                                                                                   fl_Build_cs_GNSS_AcquisitionAssistList(tsc_IntList_38_40_42_43_59_60, //@sic R5-215711 sic@
                                                                                                                                          p_RetrievedGenericAssistData[i].gnss_AcquisitionAssistance.gnss_AcquisitionAssistList),
                                                                                                   p_RetrievedGenericAssistData[i].gnss_AcquisitionAssistance.confidence_r10),
                                                                     cs_GNSS_Almanac_Bds(p_RetrievedGenericAssistData[i].gnss_Almanac),
                                                                     -,
                                                                     fl_Build_cs_GNSS_AuxiliaryInformation(tsc_IntList_3_4_5_10_18_19,  //@sic R5-241577 sic@
                                                                                                           p_RetrievedGenericAssistData[i].gnss_AuxiliaryInformation));
      //Now remove those elements not needed depending on the case
      if (not (pc_A_GPS_L1C_A or pc_A_GPS_L1C_A_MGPS)) { //Only present if GPS is also supported
        v_GenericAssistDataElement.gnss_TimeModels := omit;
      }
      //@sic R5-206424 sic@
      if (not pc_BDS_B1C or pc_BDS_B2a or pc_BDS_B3I) { // UE does not supports multiple BDS signals -> Do not include Auxiliary Information (37.571-2, Table 5.4.1.1-1) @sic R5-230336 sic@
        v_GenericAssistDataElement.gnss_AuxiliaryInformation := omit;
      }
      if (p_UEB) { // UE-Based case
        v_GenericAssistDataElement.gnss_AcquisitionAssistance := omit;
        v_GenericAssistDataElement.gnss_Almanac := omit;
      }
      else { //UE-Assisted case
        if (pc_GNSS_AA) { // UE supports GNSS Acquisition Assistance
          v_GenericAssistDataElement.gnss_NavigationModel := omit;
          v_GenericAssistDataElement.gnss_Almanac := omit;
        }
        else { // UE does not supports GNSS Acquisition Assistance
          v_GenericAssistDataElement.gnss_AcquisitionAssistance := omit;
        }
      }
    }
    return v_GenericAssistDataElement;
  } 


After change
function f_POS_SelectTemplate_GenericAssistDataElement_BDS(GNSS_GenericAssistData p_RetrievedGenericAssistData,
                                                             boolean p_UEB) return template (value) GNSS_GenericAssistDataElement
  {
    var template (value) GNSS_GenericAssistDataElement v_GenericAssistDataElement;
    var integer i := f_POS_FindIndexForGNSSMethod(p_RetrievedGenericAssistData, cs_GNSS_ID_Bds); //Indicates the position of the BDS data in the retrieved data
    
    if (i == -1) {//BDS element was not found in the list
      FatalError (__FILE__, __LINE__, "Invalid CellId"); // This function is only called if PICS indicate BDS support. If BDS assistance data is not there, it is a FatalError
    }
    else {
      //Include all possible elements in the template
      // @sic R5-240554 sic@
      v_GenericAssistDataElement := cs_GNSS_GenericAssistDataElement(cs_GNSS_ID_Bds,
                                                                     cs_GNSS_TimeModelList(tsc_GNSS_TO_ID_GPS, p_RetrievedGenericAssistData[i].gnss_TimeModels[0]),
                                                                     cs_NavigationModel_Bds(p_RetrievedGenericAssistData[i].gnss_NavigationModel),
                                                                     cs_GNSS_AcquisitionAssistance(cs_GNSS_SignalID(fl_DetermineBDS_SignalID_GNSS_AcquAssist()), // Beidou B1I or B1C
                                                                                                   fl_Build_cs_GNSS_AcquisitionAssistList(tsc_IntList_38_40_42_43_59_60, //@sic R5-215711 sic@
                                                                                                                                          p_RetrievedGenericAssistData[i].gnss_AcquisitionAssistance.gnss_AcquisitionAssistList),
                                                                                                   p_RetrievedGenericAssistData[i].gnss_AcquisitionAssistance.confidence_r10),
                                                                     cs_GNSS_Almanac_Bds(p_RetrievedGenericAssistData[i].gnss_Almanac),
                                                                     -,
                                                                     fl_Build_cs_GNSS_AuxiliaryInformation(tsc_IntList_38_40_42_43_59_60,  //@sic R5-241577 sic@
                                                                                                           p_RetrievedGenericAssistData[i].gnss_AuxiliaryInformation));
      //Now remove those elements not needed depending on the case
      if (not (pc_A_GPS_L1C_A or pc_A_GPS_L1C_A_MGPS)) { //Only present if GPS is also supported
        v_GenericAssistDataElement.gnss_TimeModels := omit;
      }
      //@sic R5-206424 sic@
      if (not pc_BDS_B1C or pc_BDS_B2a or pc_BDS_B3I) { // UE does not supports multiple BDS signals -> Do not include Auxiliary Information (37.571-2, Table 5.4.1.1-1) @sic R5-230336 sic@
        v_GenericAssistDataElement.gnss_AuxiliaryInformation := omit;
      }
      if (p_UEB) { // UE-Based case
        v_GenericAssistDataElement.gnss_AcquisitionAssistance := omit;
        v_GenericAssistDataElement.gnss_Almanac := omit;
      }
      else { //UE-Assisted case
        if (pc_GNSS_AA) { // UE supports GNSS Acquisition Assistance
          v_GenericAssistDataElement.gnss_NavigationModel := omit;
          v_GenericAssistDataElement.gnss_Almanac := omit;
        }
        else { // UE does not supports GNSS Acquisition Assistance
          v_GenericAssistDataElement.gnss_AcquisitionAssistance := omit;
        }
      }
    }
    return v_GenericAssistDataElement;
  } 
